

7. Differenziation

7.1 Definition der Ableitung


Die Ableitung einer Funktion f im Punkt x_0 ist über einen Grenzübergang definiert, bei dem die Sekante in die Tangente und damit die Sekantensteigung in die Tangentensteigung übergeht:

Definition: (Ableitung einer Funktion). Eine Funktion $f : \mathbb{D} \rightarrow \mathbb{R}$ heißt im Punkte $x_0 \in \mathbb{D}$ **differenzierbar**, falls der Grenzwert

$$f'(x_0) := \lim_{\Delta x \rightarrow 0} \frac{f(x_0 + \Delta x) - f(x_0)}{\Delta x} := \lim_{x \rightarrow x_0} \frac{f(x) - f(x_0)}{x - x_0}$$

existiert. Man bezeichnet ihn als erste **Ableitung** der Funktion f im Punkte x_0 .

Im Folgenden stellen wir diesen Grenzübergang (Sekante geht über in die Tangente) durch eine Animation am Beispiel der Funktion $f(x) = x^2$ im Punkt $x_0 = 1$ graphisch dar:



```
> f := x -> x^2: x0 := 1:
> Sekante := (f(x0 + dx) - f(x0))/dx * (x-x0) + f(x0):
> N := 10:
> for i from 1 to N
> do
>   dx := 3/i:
>   p[i] := plot ([f(x), Sekante], x = 0..4):
> end do:
> with (plots): display ([seq(p[n], n = 0..N)], insequence=true);
```



Visualisierung mit MAPLE: In der MAPLE-Animation erkennt man, dass der Punkt Q auf der Kurve von f entlang zum Punkt P wandert. Dabei nähert sich die Sekante der Tangente an und die Sekantensteigung geht in die Tangentensteigung über.

7.2 Differenzieren

MAPLE bietet zwei Möglichkeiten zur Differenziation:

`diff(ausdruck, var)` differenziert einen Ausdruck nach der Variablen `var`.

`D(funktion)` differenziert eine Funktion.

(1) Die Ableitung eines *Ausdrucks* geschieht durch den **diff**-Befehl.

> `Diff(x^2 + ln(x) + 4, x) = diff(x^2 + ln(x) + 4, x);`

$$\frac{d}{dx} (x^2 + \ln(x) + 4) = 2x + \frac{1}{x}$$

(2) Die Ableitung einer *Funktion* erfolgt mit dem **D**-Operator.

> `f := x -> exp(x) + 4 * x^2;`

> `D(f);`

$$x \rightarrow e^x + 8x$$

Das Ergebnis des **D**-Operators ist wieder eine Funktion, die anschließend an einer Stelle x_0 ausgewertet werden kann:

> `D(f)(0);`

1

⚠ Es ist wichtig zwischen **diff** und **D** zu unterscheiden: **diff** differenziert einen Ausdruck und liefert als Ergebnis einen Ausdruck; **D** differenziert eine Funktion und liefert als Ergebnis eine Funktion! Man beachte, dass gilt

(1) $D(f)(x) = \text{diff}(f(x), x)$

(2) $\text{unapply}(\text{diff}(f(x), x), x) = D(f)$

⊗ Höhere Ableitungen

Höhere Ableitungen werden in MAPLE durch den Wiederholungsoperator `$` oder `@` gebildet:

> `diff(x^2 + ln(x) + 4, x$ 2);`

$$2 - \frac{1}{x^2}$$

bzw.

```
> (D@@2)(f);
```

$$x \rightarrow e^x + 8$$

gebildet. Bei Großschreibung des **Diff**-Befehls (inerte-Form) wird die Ableitung nur symbolisch dargestellt.

7.3 Logarithmische Differenziation

7.3

Definiert man

```
> y:=x^cos(x);
```

$$y := x^{\cos(x)}$$

differenziert der **diff**-Befehl diesen Ausdruck

```
> diff(y, x);
```

$$x^{\cos(x)} \left(-\sin(x) \ln(x) + \frac{\cos(x)}{x} \right)$$

MAPLE wendet die logarithmische Differenziation also automatisch an. Man kann sie aber auch schrittweise durchführen lassen:

```
> eq:= y=x^cos(x);
```

$$eq := y = x^{\cos(x)}$$

Wir logarithmieren die Gleichung. Mit **ln(eq)** ist dies leider nicht möglich, da MAPLE dann nicht den Logarithmus der linken und rechten Seite der Gleichung berechnet

```
> ln(eq);
```

*Error, invalid input: ln expects its 1st argument, x,
to be of type algebraic, but received y = x^cos(x)*

Stattdessen wenden wir den **map**-Operator auf die Gleichung an

```
> lneq:=map(ln,eq);
```

$$lneq := \ln(y) = \ln(x^{\cos(x)})$$

⚠ Bevor nun differenziert wird, ersetzen wir y durch $y(x)$, da sonst die Ableitung von $\ln(y)$ nach x Null ergibt

```
> diff(%, x);
```

$$0 = -\sin(x) \ln(x) + \frac{\cos(x)}{x}$$

Richtig muss es lauten

```
> subs(y=y(x), lneq);
```

$$\ln(y(x)) = \ln(x^{\cos(x)})$$

> deq:=diff(%, x);

$$deq := \frac{\frac{d}{dx} y(x)}{y(x)} = -\sin(x) \ln(x) + \frac{\cos(x)}{x}$$

Obige Gleichung wird mit dem **isolate**-Befehl nach $y'(x)$ aufgelöst

> isolate(deq, diff(y(x),x));

$$\frac{d}{dx} y(x) = \left(-\sin(x) \ln(x) + \frac{\cos(x)}{x} \right) y(x)$$

7.4 Implizite Differenziation

Definiert man die Gleichung

> eq:= exp(y) - exp(2 * x) = x * y;

$$eq := e^y - e^{(2x)} = xy$$

muss vor dem Differenzieren y durch $y(x)$ ersetzt werden, da sonst die linke Seite der Gleichung differenziert Null ergibt.

> subs(y=y(x), eq):

> deq:=diff(%, x);

$$deq := \left(\frac{d}{dx} y(x) \right) e^{y(x)} - 2e^{(2x)} = y(x) + x \left(\frac{d}{dx} y(x) \right)$$

Die resultierende Gleichung nach y' aufgelöst gibt

> Diff(y(x), x) = solve(deq, diff(y(x),x));

$$\frac{d}{dx} y(x) = \frac{2e^{(2x)} + y(x)}{e^{y(x)} - x}$$

7.5 L'Hospitalsche Regeln

Die Regeln von l'Hospital werden bei MAPLE automatisch durch den **limit**-Befehl berücksichtigt:

> Limit (sin(x)/x, x = 0) = limit (sin(x)/x, x = 0);

$$\lim_{x \rightarrow 0} \frac{\sin(x)}{x} = 1$$

Auch die Fälle $0 \cdot \infty$, $\infty - \infty$, 1^∞ werden teilweise umgeformt und nach den Regeln von l'Hospital berechnet

> Limit ((1 + t * 1/x)^x, x = infinity) = limit ((1 + t * 1/x)^x, x = infinity);

$$\lim_{x \rightarrow \infty} \left(1 + \frac{t}{x} \right)^x = e^t$$

7.6 Newton-Verfahren

7.6

Das *Newton-Verfahren* ist ein schnelles numerisches Verfahren, um Nullstellen von Funktionen $f(x) = 0$ näherungsweise zu bestimmen.

Das Verfahren: Das Newton-Verfahren ist ein iteratives Verfahren: Ausgehend von einer Anfangsschätzung x_0 für die Nullstelle berechnen wir dazu in diesem Punkt x_0 die Tangente an f und bestimmen den Schnittpunkt der Tangente mit der x -Achse. Dieser Wert sei x_1 . x_1 liegt in der Regel näher an der Nullstelle als x_0 . Nun berechnet man in x_1 die Tangente der Funktion und bestimmt den Achsenschnittpunkt x_2 . Durch Fortführung des Verfahrens nähert man sich der Nullstelle an (siehe Abb. 7.1).




Abb. 7.1. Geometrische Interpretation des Newton-Verfahrens

Aufstellen der Formeln: Die Tangentengleichung in x_0 hat die Form

$$y = f(x_0) + f'(x_0)(x - x_0)$$

und der Schnittpunkt x_1 mit der x -Achse ist definiert durch $y = 0$:

$$0 = f(x_0) + f'(x_0)(x_1 - x_0).$$

Damit ist

$$x_1 = x_0 - \frac{f(x_0)}{f'(x_0)}.$$

Durch Iteration erhält man das folgende Verfahren:

Algorithmus (Newton-Verfahren)

- (1) **Initialisierung:** Wähle Startwert x_0 ; $\delta := 10^{-5}$.
- (2) **Iteration:** $x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)}$ $n = 0, 1, 2, 3, \dots$
- (3) **Abbruchbedingung:**
 - Falls $|x_{n+1} - x_n| < \delta$, dann $\xi = x_{n+1}$. Stop.
 - Falls $|x_{n+1} - x_n| \geq \delta$, dann weiter mit (2).

Falls man eine schlechte Schätzung für den Startwert hat, sollte man sich zunächst mit dem Bisektionsverfahren (\rightarrow 6.4) eine bessere Startnäherung verschaffen und anschließend das Newton-Verfahren verwenden. Oftmals kann man durch Zeichnen des Funktionsgraphen einen guten Startwert x_0 finden.

Realisierung des Newton-Verfahrens mit MAPLE: Der Algorithmus des Newton-Verfahrens wird direkt in die Prozedur **newton** übernommen. Der Aufruf der Prozedur **newton** erfolgt durch Angabe des Funktionsausdrucks und des Startwertes x_0 . Durch Verwendung des **D**-Operators in der MAPLE-Prozedur **newton** wird die Ableitung dieser Funktion explizit bestimmt.

```
> newton := proc ()
> local iter, x0, xna, xnn, delta, func, f, x;
>
> func := args[1]: x := op(1, args[2]); x0 := op(2, args[2]);
> f := unapply (func, x):
> iter := 0: delta := 1e-9:
> xna := evalf (x0):
> xnn := xna - f(xna)/D(f)(xna):
>
> while abs(xna - xnn) > delta
> do iter := iter + 1:
>   print (iter, xna, f(xna));
>   xna := xnn:
>   xnn := xna - f(xna)/D(f)(xna):
> end do;
>
> print ('Die Nullstelle liegt nach ', iter, 'Iterationen bei xi = ', xnn);
> end;
```

Beispiel 7.1 (Mit MAPLE-Worksheet). Gesucht ist eine von Null verschiedene Lösung der Gleichung

$$1 - \frac{1}{5}x = e^{-x}.$$

Um das Newton-Verfahren anwenden zu können, setzen wir

$$f(x) = 1 - \frac{1}{5}x - e^{-x}$$

und bestimmen eine Nullstelle von $f(x)$: Der Aufruf der Prozedur **newton** für das Problem erfolgt dann durch Angabe des Funktionsausdrucks $1 - \frac{1}{5}x - e^{-x}$ und des Startwertes $x_0 = 2$:

```
> y := 1 - 1/5 * x - exp(-x):
> newton (y, x = 2.);
```

n	x_n	$f(x_n)$
0	2	0.4646
1	9.1857	-.8372
2	4.9973	-.00622
3	4.9651	$-0.3 \cdot 10^{-5}$

Nach 3 Iterationen erhält man für die Lösung der Gleichung $z = 4.9651$ mit einer Genauigkeit von 4 Dezimalstellen. \square



Extras im Web: Auf der Homepage zum Buch befindet sich eine erweiterte MAPLE-Prozedur, **newton_ext**, die den Konvergenzprozess in Form einer Animation visualisiert. Die Beschreibung weiterer Verfahren befinden sich auf der Homepage im Kapitel über das Lösen von **nichtlinearen** Gleichungen.

⊙ **Quadratwurzeln**

Beispiel 7.2 (Berechnung von Wurzeln). Gesucht ist die Quadratwurzel \sqrt{a} einer positiven Zahl a . Wir interpretieren \sqrt{a} als die einzige positive Nullstelle der Funktion

$$f(x) = x^2 - a.$$

Zur numerischen Berechnung von $\sqrt{3}$ wenden wir das Newton-Verfahren auf diese Funktion an:

n	0	1	2	3	4	5
x_n	3.00000	2.00000	1.75000	1.73214	1.73205	1.73205

Die angegebene Methode ist eine der besten zur Berechnung von Quadratwurzeln. Die meisten Computerprogramme beruhen darauf. \square

⊙ **Berechnung von k -ten Wurzeln**

Bemerkung: Das Newton-Verfahren ist nicht nur auf die Berechnung von Quadratwurzeln beschränkt, sondern kann auch zur Berechnung von k -ten Wurzeln herangezogen werden. Denn $\sqrt[k]{a}$ ist die positive Nullstelle von $f(x) = x^k - a$. Mit $f'(x) = kx^{k-1}$ lautet die Newton-Folge

$$x_{n+1} = \frac{1}{k} \left[(k-1)x_n - \frac{a}{x_n^{k-1}} \right] \quad n = 0, 1, 2, 3, \dots$$

7.7 Anwendungsbeispiel: Magnetfeld von Leiterschleifen

Das durch eine stromdurchflossene Leiterschleife erzeugte Magnetfeld ist auf der Achse der Leiterschleife gegeben über die Formel

$$B(z) = \frac{\mu_0 I R^2}{2 (R^2 + z^2)^{\frac{3}{2}}},$$

wenn $R (= 0.1m)$ der Radius der Leiterschleife, $I (= 1A)$ der Strom und $\mu_0 (= 4\pi \cdot 10^{-7} H/m)$ die Permeabilität von Vakuum. Die Effekte der Stromzuleitung werden vernachlässigt.




Abb. 7.2. (a) Stromdurchflossene Leiterschleife und (b) Magnetfeld auf der Achse

Das Magnetfeld von zwei stromdurchflossenen Leiterschleifen ist die Überlagerung der Einzelmagnetfelder. Gesucht ist der Abstand d der Leiterschleifen, so dass das Magnetfeld zwischen den Einzelschleifen möglichst homogen (= gleichförmig) wird.

Um uns einen Überblick über das Magnetfeld für verschiedene Abstände d zu verschaffen, berechnen wir das Gesamtmagnetfeld

$$B = \frac{1}{2} \mu_0 I R^2 \left(\frac{1}{\left(R^2 + \left(z - \frac{1}{2}d\right)^2\right)^{3/2}} + \frac{1}{\left(R^2 + \left(z + \frac{1}{2}d\right)^2\right)^{3/2}} \right)$$

auf der Achse, wenn die erste Leiterschleife bei $z = \frac{d}{2}$ und die zweite bei $z = -\frac{d}{2}$ liegt. Das Ergebnis stellen wir als Animation dar.

```
> B:=d->mu * s/2 * R^2 * (1/(R^2
+ (z-d/2)^2)^(3/2) + 1/(R^2 +(z+d/2)^2)^(3/2));
> mu:=4 * Pi * 1e-7; s:=1.; R:=0.1;
```

⚠ Achtung: Man beachte, dass der Strom nicht mit dem Variablenname I definiert werden darf, da $I = \sqrt{-1}$ als Systemvariable vordefiniert ist!

Wir variieren den Abstand d der Spulen von $d = \frac{2}{5}R$ bis $d = 2R$ in $ndivi$ Schritten und zeichnen für jeden der Abstände das Magnetfeld auf der Achse:

```
> p:=d -> plot(B(d),z=-0.2..0.2);
```

```
p := d → plot(B(d), z = -.2...2)
```

```
> ndivi:=8;
```

```
> dr:=2 * (R - R/5) / ndivi;
```




```
> A:=seq(2 * R/5+(i-1) * dr, i=1..ndivi);
```

```
A := 0.0400000000, 0.0600000000, 0.0800000000, 0.1000000000,
0.1200000000, 0.1400000000, 0.1600000000, 0.1800000000
```

Die Animation ergibt sich aus

```
> with(plots):
```

```
> display([seq(p(i), i=A)], insequence=true, axes=framed, thickness=2);
```



Das zugehörige MAPLE-Worksheet erzeugt die obigen Bilder als Animation, bei der der Abstand d zwischen 0.04 und 0.18 variiert. Es zeigt sich, dass bei $d = 0.1$ das Magnetfeld relativ homogen verläuft. Zur Skalierung wurde das Magnetfeld in $10^{-6} [T]$ angegeben.

Die mathematische Bedingung für die Homogenität des Magnetfeldes bei $z = 0$ ist, dass $B''(z = 0) = 0$.

> mu:='mu': s:='s': R:='R':
> diff(B(d), z\$2);

$$\frac{1}{2}\mu i R^2 \left(\frac{15}{4} \frac{(2z-d)^2}{\left(R^2 + \left(z - \frac{1}{2}d\right)^2\right)^{7/2}} - 3 \frac{1}{\left(R^2 + \left(z - \frac{1}{2}d\right)^2\right)^{5/2}} \right. \\ \left. + \frac{15}{4} \frac{(2z+d)^2}{\left(R^2 + \left(z + \frac{1}{2}d\right)^2\right)^{7/2}} - 3 \frac{1}{\left(R^2 + \left(z + \frac{1}{2}d\right)^2\right)^{5/2}} \right)$$

Wir setzen obigen Ausdruck für $B(d)$ bei $z = 0$ Null und lösen nach d auf

> d_homo:= solve (subs(z=0,%) = 0, d);

$$d_homo := R, -R$$

Ist der Spulenabstand d gleich dem Radius R der Spulen, so ist das Magnetfeld auf der Achse homogen (*Helmholtz-Spulen*)!

7.8

7.8 Zusammenstellung der MAPLE-Befehle

Differenziations-Befehle von MAPLE

diff (y, x)	Ableitung des Ausdrucks y nach x .
diff (y, x \$ n)	n -te Ableitung des Ausdrucks y nach x .
Diff (y, x)	Symbolische Darstellung der Ableitung.
D (f)	Ableitung der Funktion f .
(D@@n) (f)	n -te Ableitung der Funktion f .

MAPLE-Worksheets zu Kapitel 7



Die folgenden elektronischen Arbeitsblätter stehen für Kapitel 7 mit MAPLE zur Verfügung.

- Zum Ableitungsbegriff
- Differenzieren mit MAPLE
- Kurvendiskussion mit MAPLE
- Helmholtz-Spulen mit MAPLE
- Newton-Verfahren mit MAPLE